ALAN HALLECK'S RAZER1
and
DR. PANKNIN'S TWIST FORMULAE

This month we describe a computer program to help design swept 'wings, and the result of Alan Halleck's use of the program - his RAZER1 slope racer!

Our trip to the 1989 MARCS Symposium was a wonderful experience, and we wrote about Dr. Walter Panknin's presentation on flying wings, "Flying Rainbows," in the September 1990 issue of RCSD. As we mentioned in that report, Walter gave out a packet of materials to those interested in designing their own flying wings. Included were the formulae for determing wing twist based on required C_L and a stability factor.

Using Walter's formulae as a basis, we developed a short computer program for our antiquated Apple II Plus. Written to compute needed wing twist, it ran very rapidly and gave twist values comparable to several known successful designs.

Alan Halleck, of Portland, Oregon, is a fellow "'wing nut" and computer freak with whom we converse on a regular basis. Knowing Alan would be interested, we sent down a hard copy of this small program for him to enter into his IBM compatible. Since there are no graphics involved and the commands used are parallel for both versions of BASIC he was able to enter it with no problem, and Alan immediately set to work designing a flying wing using a couple of airfoils designed by Martin Hepperle.

An October '90 get together with Alan during which we spent several hours at his computer produced a much more sophisticated BASIC program. The program now prompts the user for airframe information such as span, chords and airfoils, sweep, projected weight, and other information. The printout shows the wing area, location of the neutral point, CG location based on a series of stability factors, wing loading, and of course the twist required for a given coefficient of

lift. It's easy to modify individual pieces of data to see the effect as all values remain constant from one run to another unless changed when prompted. All airfoil data required by the program is stored on disk, and placing airfoil data on disk is a simple task performed by a very small additional program.

The result of Alan's design work is a bat-tailed 20° sweep flying wing of 77" span. Winglets provide some vertical area, but elevons are the only control surfaces. Total washout is a minimal 1 1/2 degrees. The 'wing is of foam core construction with fiberglass and Kevlar providing the strength, and the entire structure is vacuum bagged. We've included a small sketch of the resulting planform. Alan located the CG according to the computer program and found it to be extremely accurate. Removal of weight to shift the CG rearward was met with a decrease in performance. This is proof again the program does deliver accurate information.

We had the opportunity of witnessing the RAZER1 in action at Goodnoe Hills on the Columbia River. During a high speed landing the 'wing flipped into the air and hit the rock surface of the hill inverted but survived without a scratch - it's one strong airplane! The RAZER1 is capable of some great aerobatics. It does good axial rolls and can fly inverted for extended periods. The turning radius is very small.

The proof of the RAZER1 design came at the November '90 slope race held by Alan's local club, the Portland Area Sailplane Society (PASS). In winds of 40 knots and above the RAZER1 performed admirably, taking second place in all heats and placing fifth overall out of 16 entries. Alan admits to not being well practiced for the event, and he missed a pylon on the last lap of the last race; otherwise his placing would have been higher.

New design, first race, pretty good performance, right? Well, there's more. The wing loading of the RAZER1 is about 10.7 oz. per sq. ft., yet it was competing against conventional tailed sailplanes loaded at 16 to 24 oz. per sq. ft. The rotor on the

hilltop was viscious and ate several airplanes, but the RAZER1's single "hard landing" barely dented the nose.

We and Alan wish to Dr. Panknin for presenting his formulae to the modeling public at the '89 MARCS Symposium, as well as for so enthusiastically supporting the release of our computer program to readers of RCSD. The complete text of Walter's presentation at the '89 Symposium is within the Proceedings available from MARCS.



APPLESOFT VERSION

INTRODUCTION TO THE PANKNIN TWIST PROGRAM:

This first program asks for certain data about your preliminary design. Initially, all will have zero values. Fill in data as requested. When complete, the program will print the input information along with a series of derived parameters. Following that, a sequence of stability factors, then the twist required, and the location of the CG for each stability factor. During the second and subsequent run throughs, the data originally input will be repeated on the screen. If you wish to retain this data, simply press <RETURN>. If a change is needed input the new data and then press <RETURN>. In this way the program can go through a series of single or multiple parameter changes, giving relevent information for each iteration.

```
10  TEXT : HOME :PO = 2
20  VTAB 10: PRINT "                PANKNIN.TWIST"
30  FOR X = 0 TO 5000: NEXT X
42  HOME
45  ONERR  GOTO 5000
50  PRINT "PLANE TYPE OR NAME ------ = ";AC$: INPUT Z$: IF Z$ <  >
"" THEN AC$ = Z$
51  PRINT  CHR$ (4);"OPEN";AC$
52  PRINT  CHR$ (4);"READ";AC$
53  INPUT AC$
54  INPUT A$
55  INPUT B$
56  INPUT LW
57  INPUT LA
58  INPUT B
59  INPUT PF
60  INPUT CA
61  INPUT W
70  PRINT  CHR$ (4);"CLOSE";AC$
78  PRINT "AIRFOIL NAME ROOT ----- = ";A$;"  ";: INPUT Z$: IF Z$ <
> "" THEN A$ = Z$
79  PRINT "               TIP  ---- = ";B$;"  ";: INPUT Z$: IF Z$ <
> "" THEN B$ = Z$
80  PRINT "ROOT CHORD ------------- = ";LW;"  ";
100  INPUT Z$: IF Z$ <  > "" THEN LW =  VAL (Z$)
110  PRINT "TIP CHORD -------------- = ";LA;"  ";
130  INPUT Z$: IF Z$ <  > "" THEN LA =  VAL (Z$)
140  PRINT "SPAN      -------------- = ";B;"  ";
160  INPUT Z$: IF Z$ <  > "" THEN B =  VAL (Z$)
170  PRINT "SWEEPBACK ANGLE OF 1/4 CHORD LINE = ";PF;"  ";
190  INPUT Z$: IF Z$ <  > "" THEN PF =  VAL (Z$)
200  PRINT "CALCULATE WITH CL = ";CA;"  ";
220  INPUT Z$: IF Z$ <  > "" THEN CA =  VAL (Z$)
230  PRINT "WEIGHT IN OUNCES = ";W;"  ";
250  INPUT Z$: IF Z$ <  > "" THEN W =  VAL (Z$)
260  PRINT  CHR$ (4);"OPEN";A$
265  PRINT  CHR$ (4);"READ";A$
270  INPUT NR$,CL(1),CD(1),OW,MW,TH(1),CR
280  PRINT  CHR$ (4);"CLOSE";A$
```

```
290  PRINT  CHR$ (4);"OPEN";B$
295  PRINT  CHR$ (4);"READ";B$
300  INPUT NT$,CL(2),CD(2),OA,MA,TH(1),CT
310  PRINT  CHR$ (4);"CLOSE";B$
320 TM = (LW + LA) / 2:AR = B / TM
330 TR = LA / LW
340  PRINT : PRINT
350  PRINT "IF YOU WISH A PRINTOUT OF THE DATA, PRESS <P>": PRINT
"THEN <ENTER> NOW!": PRINT : PRINT "FOR NO PRINTOUT, PRESS
<ENTER> ALONE."
360  INPUT Z$: PRINT Z$: IF Z$ = "P" THEN  PRINT  CHR$
(4)"PR#";PO: PRINT  CHR$ (9);"I80N" : REM  <CNTRL> <I> to set
printer to 80 columns
370  HOME
380  PRINT "AIRCRAFT NAME: ";AC$: PRINT "ROOT SECTION = ";A$:
PRINT "TIP SECTION = ";B$
390  PRINT "ROOT CHORD = ";LW: PRINT "TIP CHORD = ";LA
400  PRINT "SPAN = ";B: PRINT "SWEEP ANGLE OF 1/4 CHORD LINE =
";PF: PRINT "WEIGHT = ";W;" OUNCES": PRINT "CALCULATED FOR CL =
";CA
410  PRINT : PRINT
420  PRINT "AVERAGE CHORD = ";TM: PRINT "TAPER RATIO = ";TR
430  PRINT "ASPECT RATIO = ";AR
440 K1 = 1 / 4 * (3 + 2 * TR + TR ^ 2) / (1 + TR + TR ^ 2)
450 K2 = 1 - K1
460 B1 = B / 2
470 D =  TAN (PF * 3.1414927 / 180) * B1
480 L1 = LW * .25
490 L2 = LA * .25
500 D1 = (L1 - L2) + D
510 AC = (LW ^ 2 + (LW * LA) + LA ^ 2) / (6 * (LW + LA)) + (((2 *
LA) + LW) * D1) / (3 * (LW + LA))
520  PRINT "AERO CENTER = ";AC;" INCHES": PRINT "BEHIND LEADING
EDGE AT ROOT"
530  PRINT : PRINT "D1 (DISTANCE LE WINGTIP BEHIND LE ROOT)":
PRINT "= ";D1: PRINT
540  PRINT "WING AREA = ";B * TM;"SQ. IN.": PRINT  TAB( 13)B * TM
/ 144;"SQ. FT."
550  PRINT "WING LOADING = ";W / (B * TM / 144);" OZ./SQ. FT."
560  INPUT Z$: PRINT Z$: HOME
570  FOR ST = .01 TO .03 STEP .005: GOSUB 610: NEXT ST
580  INPUT Z$: PRINT Z$: HOME
590  FOR ST = .035 TO .051 STEP .005: GOSUB 610: NEXT ST
600  PRINT  CHR$ (4);"PR#0": INPUT Z$: PRINT Z$
605  PRINT "DO YOU WANT THE RAW DATA SAVED TO DISK? (Y/N)";:
INPUT Z$: IF Z$ = "Y" THEN  GOTO 900
607  PRINT "ANOTHER? (Y/N)";: INPUT Z$: IF Z$ = "Y" THEN  GOTO
50
608  END
610 TWIST = ((K1 * MW + K2 * MA) - CA * ST) / (.000014 * AR ^
1.43 * PF)
620  PRINT "ST = ";ST;"    AERO TWIST = ";TWIST
630 TWIST = TWIST - (OW - OA)
640  PRINT "              GEO TWIST = ";TWIST
650 CG = AC - TM * ST
660  PRINT "     CG = ";CG: PRINT "     FROM LE AT ROOT"
670  PRINT
```

```
680  RETURN
900  REM  DISK SAVE
910  HOME : PRINT "THIS AIRCRAFT IS CURRENTLY NAMED ";AC$
920  PRINT : PRINT "IF THIS IS OK, PRESS <ENTER>, OTHERWISE ENTER
THE NEW NAME"
930  INPUT Z$: IF Z$ <  > "" THEN AC$ = Z$
950  HOME : PRINT "SAVING TO DISK..."
1020  PRINT  CHR$ (4)"OPEN";AC$
1022  PRINT  CHR$ (4)"DELETE";AC$
1024  PRINT  CHR$ (4)"OPEN";AC$
1026  PRINT  CHR$ (4)"WRITE";AC$
1100  PRINT AC$
1110  PRINT A$
1120  PRINT B$
1130  PRINT LW
1140  PRINT LA
1150  PRINT B
1160  PRINT PF
1161  PRINT CA
1162  PRINT W
1170  PRINT  CHR$ (4)"CLOSE";AC$
1180  GOTO 607
5000 E1 =  PEEK (222): PRINT "ERROR #";E1: IF E1 = 5 THEN  PRINT
"NO FILE BY THAT NAME": GOTO 78
5001  END
```

INTRODUCTION TO THE AIRFOIL DATA PROGRAM:

This small program places input airfoil information on disk for use by the above program. Other programs, of your own design perhaps, can be written to make use of the information as well. The program runs in a self explanatory manner.

```
10  TEXT : HOME
20  PRINT "AIRFOIL NAME = ": PRINT "ENTER 'DONE' IF NO MORE ";:
INPUT N$
30  IF N$ = "DONE" THEN  END
40  INPUT "CL = ";CL
50  INPUT "CD = ";CD
60  INPUT "ZERO LIFT ANGLE = ";ZLA
70  INPUT "PITCHING MOMENT = ";CM
80  INPUT "THICKNESS = ";TH
90  INPUT "CAMBER = ";C
100  PRINT  CHR$ (4)"OPEN";N$
110  PRINT  CHR$ (4)"DELETE";N$
120  PRINT  CHR$ (4)"OPEN";N$
130  PRINT  CHR$ (4)"WRITE";N$
140  PRINT N$
150  PRINT CL
160  PRINT CD
170  PRINT ZLA
180  PRINT CM
190  PRINT TH
200  PRINT C
210  PRINT  CHR$ (4)"CLOSE";N$
220  GOTO 10
```

FOR IBM and IBM COMPATIBLES

INTRODUCTION TO THE PANKNIN TWIST PROGRAM:

This first program asks for certain data about your preliminary design. Initially, all will have zero values. Fill in data as requested. When complete, the program will print the input information along with a series of derived parameters. Following that, a sequence of stability factors, then the twist required, and the location of the CG for each stability factor. During the second and subsequent run throughs, the data originally input will be repeated on the screen. If you wish to retain this data, simply press <RETURN>. If a change is needed input the new data and then press <RETURN>. In this way the program can go through a series of single or multiple parameter changes, giving relevent information for each iteration.

```
10  CLS
20  PRINT "                          **********   PANKNIN TWIST
**********"
30  FOR X = 0 TO 5000 : NEXT X
45  ON ERROR GOTO 5000
50  CLS : PRINT "PLANE TYPE OR NAME ------ = "; AC$:  INPUT Z$: IF
Z$<>"" THEN AC$ = Z$
52  OPEN "I",#1,AC$
53  INPUT #1,AC$
54  INPUT #1,A$
55  INPUT #1,B$
56  INPUT #1,LW
57  INPUT #1,LA
58  INPUT #1,B
59  INPUT #1,PF
60  INPUT #1,CA
61  INPUT #1,W
70  CLOSE #1
78  PRINT "AIRFOIL NAME  ROOT ----- = ";A$: INPUT Z$:IF Z$<>""
THEN A$ = Z$
79  PRINT "               TIP  ----- = ";B$: INPUT Z$:IF Z$<>""
THEN B$ = Z$
80  PRINT "ROOT CHORD ---------------- = ";: IF LW = 0 THEN GOTO 100
90  PRINT LW
100  INPUT Z$:IF Z$<>"" THEN LW = VAL(Z$)
110  PRINT "TIP CHORD ---------------- = ";:IF LA = 0 THEN GOTO 130
120  PRINT LA
130  INPUT Z$:IF Z$<>"" THEN LA = VAL(Z$)
140  PRINT "SPAN       ---------------- = ";:IF B = 0 THEN GOTO 160
150  PRINT B
160  INPUT Z$:IF Z$<>"" THEN B = VAL (Z$)
170  PRINT "SWEEPBACK ANGLE OF 1/4 CHORD LINE = ";:IF PF = 0 THEN
GOTO 190
180  PRINT PF
190  INPUT Z$:IF Z$<>"" THEN PF = VAL(Z$)
200  PRINT "CALCULATE WITH CL = ";:IF CA = 0 THEN GOTO 220
210  PRINT Z$
```

```
220  INPUT Z$:IF Z$<>"" THEN CA = VAL(Z$)
230  PRINT "WEIGHT IN OUNCES = ";:IF W= 0 THEN GOTO 250
240  PRINT W
250  INPUT Z$:IF Z$<>"" THEN W = VAL(Z$)
260  OPEN "I",#1,A$
270  INPUT #1,NR$,CLR,CDR,OW,MW,THR,CR
280  CLOSE #1
290  OPEN #1,#1,B$
300  INPUT #1,NT$,CLT,CDT,OA,MA,THT,CT
310  CLOSE #1
320  TM = (LW + LA) / 2:AR = B / TM
330  TR = LA / LW
340  PRINT : PRINT
350  PRINT "IF YOU WISH A PRINTOUT OF THE DATA, PRESS <CTRL> AND
<PRINT SCREEN>," : PRINT "THEN <ENTER> NOW!": PRINT : PRINT "FOR
NO PRINTOUT, PRESS <ENTER> ALONE."
360  INPUT Z$: PRINT Z$
370  CLS
380  PRINT "AIRCRAFT NAME: "; AC$: PRINT "ROOT SECTION = "; A$:
PRINT "TIP SECTION = ";B$
390  PRINT "ROOT CHORD = ";LW: PRINT "TIP CHORD = ";LA
400  PRINT "SPAN = ";B: PRINT "SWEEP ANGLE OF 1/4 CHORD LINE =
";PF: PRINT "WEIGHT = ";W;" OUNCES": PRINT "CALCULATED FOR CL =
";CA
410  PRINT : PRINT
420  PRINT "AVERAGE CHORD = ";TM: PRINT "TAPER RATIO = ";TR
430  PRINT "ASPECT RATIO = ";AR
440  K1 = 1/4 * (3 + 2 * TR + TR ^ 2)/(1 + TR + TR ^ 2)
450  K2 = 1 - K1
460  B1 = B / 2
470  D = TAN (PF * 3.1414927 / 180) * B1
480  LW1 = LW * .25
490  LA1 = LA = .25
500  D1 = (LW1 - LA1) + D
510  AC = (LW ^ 2 + (LW * LA) + LA ^ 2) / (6 * (LW + LA)) + (((2
* LA) + LW) * D1) / (3 * (LW + LA))
520  PRINT "AERODYNAMIC CENTER = ";AC;" INCHES BEHIND LEADING
EDGE AT ROOT"
530  PRINT : PRINT "D1 (DISTANCE LEADING EDGE OF WINGTIP": PRINT
"IS BEHIND LEADING EDGE OF ROOT) = ";D1: PRINT
540  PRINT "WING AREA = ";B * TM;"SQ. IN.": PRINT TAB(13) B * TM
/ 144;"SQ. FT."
550  PRINT "WING LOADING = ";W / (B * TM / 144);" OZ./SQ. FT."
560  INPUT Z$: PRINT Z$ : CLS
570  FOR ST = .01 TO .03 STEP .005: GOSUB 610 : NEXT ST
580  INPUT Z$: PRINT Z$ : CLS
590  FOR ST = .035 TO .051 STEP .005: GOSUB 610 NEXT ST
600  INPUT Z$: PRINT Z$
605  PRINT "DO YOU WANT THE RAW DATA SAVED TO DISK? (Y/N)";:INPUT
Z$:IF Z$ = "Y" THEN GOTO 900
607  PRINT "ANOTHER? (Y/N)";:INPUT Z$: IF Z$ = "Y" THEN GOTO 50
608  END
610  TWIST = ((K1 * MW + K2 * MA) - CA * ST) / (.000014 * AR ^
1.43 * PF)
620  PRINT "ST = ";ST; "    AERODYNAMIC TWIST REQ'D = ";TWIST
```

```
630  TWIST = TWIST - (OW - OA)
640  PRINT"              GEOMETRIC TWIST REQ'D = ";TWIST
650  CG = AC - TM * ST
660  PRINT"              CG = ";CG;" BACK FROM LEADING EDGE AT
ROOT"
670  PRINT
680  RETURN
900  REM DISK SAVE
910  CLS : PRINT "THIS AIRCRAFT IS CURRENTLY NAMED ";AC$
920  PRINT : PRINT "IF THIS IS OK, PRESS <ENTER>, OTHERWISE ENTER
THE NEW NAME"
930  INPUT Z$: IF Z$<>"" THEN AC$ = Z$
950  CLS : PRINT "SAVING TO DISK..."
1020  OPEN "O",#1,AC$
1100  PRINT #1,AC$
1110  PRINT #1,A$
1120  PRINT #1,B$
1130  PRINT #1,LW
1140  PRINT #1,LA
1150  PRINT #1,B
1160  PRINT #1,PF
1161  PRINT #1,CA
1162  PRINT #1,W
1170  CLOSE #1
1180  GOTO 607
5000  RESUME 78
```

INTRODUCTION TO THE AIRFOIL DATA PROGRAM:

This small program places input airfoil information on disk for use by the above program. Other programs, of your own design perhaps, can be written to make use of the information as well. The program runs in a self explanatory manner.

```
5  PRINT "AIRFOIL NAME = ": PRINT "ENTER 'DONE' IF NO MORE ";:
INPUT N$
10  IF N$ = "DONE" THEN END
40  INPUT "CL = ";CL
50  INPUT "CD = ";CD
60  INPUT "ZERO LIFT ANGLE = ";ZLA
70  INPUT "PITCHING MOMENT = ";CM
80  INPUT "THICKNESS = ";TH
90  INPUT "CAMBER = ";C
95  OPEN "O",#1,N$
100  PRINT #1,N$
110  PRINT #1,CL
120  PRINT #1,CD
130  PRINT #1,ZLA
140  PRINT #1,CM
150  PRINT #1,TH
160  PRINT #1,C
170  CLOSE #1
180  GOTO 5
```